

Making regression tables simplified

Ben Jann
ETH Zurich

Abstract. `estout`, introduced by Jann (2005), is a useful tool for producing regression tables from stored estimates. However, its syntax is relatively complex and commands may turn out lengthy even for simple tables. Furthermore, having to store the estimates beforehand can be a bit cumbersome. To facilitate the production of regression tables, I therefore present here two new commands called `eststo` and `esttab`. `eststo` is a wrapper for official Stata’s `estimates store` and simplifies the storing of estimation results for tabulation. `esttab`, on the other hand, is a wrapper for `estout` and simplifies compiling nice-looking tables from the stored estimates without much typing. In addition, updates to `estout` and `estadd` are provided.

Keywords: `st0001`, `csv`, `estadd`, `estimates`, `estout`, `eststo`, `esttab`, `excel`, `html`, `latex`, regression table, `rtf`, `word`

1 Introduction

Statistical software packages such as Stata provide a rich variety of statistical estimation routines, all producing some kind of output. While the outputs from single commands are important for scientific research, they are usually not well designed for presentation. It is often necessary to select and rearrange results from different outputs to get an overview of the results and to present a clear and concise analysis. Therefore, not only statistical routines are necessary, but also tools to efficiently processing results for presentation.

Jann (2005) provided such a tool called `estout`. `estout` compiles regression tables containing results from one or more estimation commands for use in, say, $\text{\LaTeX}$ documents, spreadsheet programs, or word processors. The initial motivation behind `estout` was to provide a generic program to compile a table from several sets of regression estimates and write the table to disk for subsequent use with other software. Developmental efforts were directed more towards functionality—to be able to flexibly arrange and format the table—than towards ease-of-use. Furthermore, since there are different needs and conventions concerning the contents and look of a regression table, the basic approach was to provide a “clean desk” for users from which they could start building-up their fully-fledged end-product. “Clean desk” means here that `estout` was designed to produce a plain, essentially unformatted table containing only point estimates by default.

Although `estout` is quite powerful in terms of functionality, the motivational orientation outlined above brought with it some limitations. These limitations may be summarized as follows.

1. **estout** tables are usually not suitable for display in Stata's results window. For example, by default, **estout** uses the tab-character to separate the table's columns. However, tab-characters are expanded to a fixed amount of blanks in the results window, causing the table's columns to appear misaligned. This is unfavorable because it is often purposeful to produce regression tables on the fly, for quick results inspection on screen.
2. **estout**'s syntax is not as intuitive and user-friendly as it could be. For example, there are nested options, which do their job, but are hard to handle and hard to remember. Also even experienced users are often forced to consult the command's documentation while producing an **estout** table.
3. The amount of typing required to compile even a simple table can be quite considerable. Users generally have to specify many options to determine the content and formatting of the table according to their needs. Note that **estout** provides the possibility to pre-specify options via so-called defaults files (similar to scheme files for Stata graphics). However, maintaining such defaults files does not appear beneficial unless one is working on large reports containing lots of similar tables.

A consequence of these limitations is that the use of **estout** may be somewhat clumsy in daily work. In addition, smooth application of **estout** is compromised by the fact that the estimation sets have to be stored using official Stata's **estimates store** before they can be tabulated (at least if there is more than one set of estimates). A drawback of **estimates store** is that it requires the user to specify names under which to store the estimation sets. Having to provide such names, although sensible in some contexts, can be distracting.

To summarize, there seems to be a need for (1) an easy-to-use version of **estout** that produces fully formatted tables right away and is suitable for interactive work, and (2) a simplified procedure to hold on to estimates for tabulation. In the remainder of this text I will address these two points (in reverse order) and present in Section 2 a command called **eststo** to overcome the limitations of **estimates store**. Section 3 then introduces a user-friendly **estout** wrapper called **esttab** and illustrates its application by examples. The Appendix (Section 4) contains syntax overviews for the two commands and provides updates to **estout** and **estadd**.

Space limitations do not allow an extended treatment of the new commands. For details and additional examples consult the online help or visit the **estout** web site at <http://fmwww.bc.edu/repec/bocode/e/estout/>.

2 **eststo**: Storing estimates simplified

The new **eststo** command stores a copy of the active estimation results for later tabulation. It is an alternative to official Stata's **estimates store**. The basic syntax of **eststo** is

```
eststo [ name ] [ , options ] [ : estimation command ]
```

Store estimates without providing a name

A main advantage of `eststo` over `estimates store` is that it does not require the user to specify a name for the stored estimates. If *name* is provided, then, naturally, the estimates are stored under this name. However, if no name is provided, `eststo` makes up its own name. Note that `eststo` keeps track of the names of the stored estimation sets via global macros from where the names can be picked up by, say, `estout`. Here is an example:

```
. sysuse auto
(1978 Automobile Data)
. regress price weight mpg
(output omitted)
. eststo
(est1 stored)
. regress price weight mpg foreign
(output omitted)
. eststo
(est2 stored)
. estout, style(fixed)
```

	est1	est2
	b	b
weight	1.746559	3.464706
mpg	-49.51222	21.8536
foreign		3673.06
_cons	1946.069	-5853.696

To erase the estimation sets that have been stored by `eststo`, type

```
. eststo clear
```

Use eststo as a prefix command

As is illustrated in the example above, a model's results are stored by applying `eststo` after the model has been fitted. This is also how official `estimates store` works. Alternatively, however, `eststo` may be used as a prefix command (see [U] 11.1.10 **Prefix commands**). For example:

```
. eststo: regress price weight mpg
(output omitted)
. eststo: regress price weight mpg foreign
(output omitted)
. estout, style(fixed)
```

	est1	est2
	b	b
weight	1.746559	3.464706
mpg	-49.51222	21.8536

```
foreign          3673.06
_cons          1946.069 -5853.696
```

Note that `by` (see [D] `by`) is allowed with `eststo`, if `eststo` is used as a prefix command. Furthermore, note that `eststo` has an option to drop the `e(sample)` from the stored estimation sets to preserve memory.¹

3 `esttab`: Tabulating estimates simplified

The new `esttab` command is a wrapper for `estout`. Its syntax is much simpler than that of `estout` and, by default, it produces publication-style tables that display nicely in Stata's results window. Note that `esttab` is more than just a simplified version of `estout`. On the one hand, `esttab` provides full `estout` functionality since all `estout` options are, in fact, allowed in `esttab`. On the other hand, `esttab` also extends functionality. For example, `esttab` adds support for Word RTF and Excel CSV and provides improved functionality for L^AT_EX and HTML.

In what follows I will try to give a brief introduction to `esttab` and illustrate some of its applications, although I will only be able to scratch the surface. See the Appendix for a syntax overview and consult the online help for more detailed information. Further examples can also be found at <http://fmwww.bc.edu/repec/bocode/e/estout/>.

Basic syntax and usage

The syntax of `esttab` is

```
esttab [ namelist ] [ using filename ] [ , options estout_options ]
```

where *namelist* is a list of names of stored estimation sets. *namelist* may be `*` to tabulate all stored estimation sets. If *namelist* is omitted, `esttab` tabulates the currently active estimates or, if present, the estimation sets stored by `eststo`. Specifying `using` causes the regression table to be written to a file on disk instead of being displayed in Stata's results window.

Analogous to `estout` (or official Stata's `estimates table`), the basic procedure is to first store a number of models and then apply `esttab` to these stored estimation sets to compose a regression table. The great difference between `esttab` and `estout`, however, is that, if applied without any options, `esttab` produces a fully formatted table. Consider the following example and compare it to the examples in Section 2, in

1. Stored estimates consume a considerable amount of memory. In order to preserve full functionality of postestimation commands (see [U] **20 Estimation and postestimation commands**), an estimation sample indicator variable (i.e. a copy of the `e(sample)` function) is stored for each estimation set. Even though byte storage type is used for these variables, they may blow up the dataset if it contains a large number of observations or if many estimation sets are stored. Additionally, storing the `e(sample)` information has the side effect of slowing down cycling through the stored estimation sets in large datasets, which also slows down tabulation programs such as `estout` or official Stata's `estimates table`.

which `estout` was used:

```
. sysuse auto
(1978 Automobile Data)
. eststo: regress price weight mpg
(output omitted)
. eststo: regress price weight mpg foreign
(output omitted)
. esttab
```

	(1) price	(2) price
weight	1.747** (2.72)	3.465*** (5.49)
mpg	-49.51 (-0.57)	21.85 (0.29)
foreign		3673.1*** (5.37)
_cons	1946.1 (0.54)	-5853.7 (-1.73)
N	74	74

t statistics in parentheses
 * p<0.05, ** p<0.01, *** p<0.001

Change table contents and add summary statistics

The default in `esttab` is to display raw point estimates along with t statistics and to print the number of observations in the table footer. Furthermore, stars denoting the significance of coefficients are displayed. All this can be changed. To replace the t statistics by standard errors, add the adjusted R -squared, and omit the significance stars, for example, type:

```
. esttab, se ar2 nostar
```

	(1) price	(2) price
weight	1.747 (0.641)	3.465 (0.631)
mpg	-49.51 (86.16)	21.85 (74.22)
foreign		3673.1 (684.0)
_cons	1946.1 (3597.0)	-5853.7 (3377.0)
N	74	74
adj. R-sq	0.273	0.478

Standard errors in parentheses

The t statistics can also be replaced by p -values (`p`), confidence intervals (`ci`), or any parameter statistics contained in the estimates (see the `aux()` option). Further summary statistics options are, for example, `pr2` for the pseudo R -squared and `bic` for Schwarz's information criterion. Moreover, there is a generic `scalars()` option to include any other scalar statistics contained in the stored estimates. For instance, `scalars(F df_m df_r)` would add the overall F statistic and information on the degrees of freedom.

Also the point estimates may be replaced by other statistics. Here is an example in which beta coefficients are printed and the t statistics are suppressed:

```
. esttab, beta not
```

	(1) price	(2) price
weight	0.460**	0.913***
mpg	-0.097	0.043
foreign		0.573***
N	74	74

Standardized beta coefficients
* p<0.05, ** p<0.01, *** p<0.001

Further possibilities are provided by the `main()` option (replace the point estimates by any other stored parameter statistics), or by `estout` options such as `eform` or `margin`.

Layout, labels, and titles

There are many options for changing the table's design and adding labels, titles, or notes. For example, the `wide` option arranges point estimates and t statistics beside one another instead of beneath one another:

```
. esttab, wide
```

	(1) price	(2) price		
weight	1.747**	(2.72)	3.465***	(5.49)
mpg	-49.51	(-0.57)	21.85	(0.29)
foreign			3673.1***	(5.37)
_cons	1946.1	(0.54)	-5853.7	(-1.73)
N	74		74	

t statistics in parentheses
* p<0.05, ** p<0.01, *** p<0.001

Furthermore, the `plain` option produces a minimally formatted table with all display formats set to Stata's %9.0g quasi-standard, and `compress` reduces horizontal spacing

to fit more models on screen without line breaking. Other options are, for example, `label` to cause variable labels to be used and `mtitles()` to specify model titles to be printed in the header above the model columns. Example:

```
. esttab, se ar2 nostar brackets label title(This is a regression table)
> nonumbers mtitles("Model A" "Model B") addnote("Source: auto.dta")
This is a regression table
```

	Model A	Model B
Weight (lbs.)	1.747 [0.641]	3.465 [0.631]
Mileage (mpg)	-49.51 [86.16]	21.85 [74.22]
Car type		3673.1 [684.0]
Constant	1946.1 [3597.0]	-5853.7 [3377.0]
Observations	74	74
Adjusted R-squared	0.273	0.478

Standard errors in brackets
Source: auto.dta

Numerical formats

`esttab` has sensible default settings for numerical display formats. For example, t statistics are printed using two decimal places and R -squared measures are printed using three decimal places. For point estimates and, for example, standard errors an adaptive display format is used where the number of displayed decimal places depends on the scale of the statistic to be printed (the default format is `a3`; see below).

The format applied to a certain statistic can be changed by adding the appropriate display format specification in parentheses. For example, to display p -values and the R -squared using four decimal places and display point estimates using the `%9.0g` format, type

```
. esttab, b(%9.0g) p(4) r2(4) nostar wide
```

	(1) price		(2) price	
weight	1.746559	(0.0081)	3.464706	(0.0000)
mpg	-49.51222	(0.5673)	21.8536	(0.7693)
foreign			3673.06	(0.0000)
_cons	1946.069	(0.5902)	-5853.696	(0.0874)
N	74		74	
R-sq	0.2934		0.4996	

p -values in parentheses

Available formats are official Stata’s display formats, such as `%9.0g` or `%3.2f` (see [D] **format**). Alternatively, as is illustrated in the example above, a fixed format can be requested by specifying a single integer indicating the desired number of decimal places. Furthermore, an adaptive format, `a#`, may be specified, where `#` in $\{1, \dots, 9\}$ determines the minimum number of “significant digits” to be printed.

Document formats

Output format options in `esttab` allow the user to quickly switch between different document formats depending on the table’s intended use. Available formats are:

`smcl` to produce an SMCL formatted table. `smcl` is the default (unless `using` is specified) and is used to display the table in Stata’s results window.

`fixed` to produce a fixed-format ASCII table. This is suitable, for example, if the table is to be displayed in a fixed-font text editor.

`tab` to produce a tab-delimited ASCII table. This is a general format that can be used as an input format for many programs.

`csv` to produce a CSV (Comma Separated Value format) table for use with Excel. Delimiter is a comma. See below for details on using `esttab` with Excel.

`scsv` to produce a CSV table using a semicolon as the delimiter. This is appropriate for some non-English versions of Excel such as the German version.

`rtf` to produce a Rich Text Format table for use with word processors. The code follows the guidelines in Burke (2003) and should work with about any RTF viewer. See below for details on using `esttab` with Word.

`html` to produce a simple HTML-formatted table that can be displayed in a web browser.

`tex` to produce a table to be included in a $\text{\LaTeX}$ 2 ϵ document. See below for details on using `esttab` with $\text{\LaTeX}$.

`booktabs` to produce a $\text{\LaTeX}$ 2 ϵ -formatted table for use with $\text{\LaTeX}$ ’s `booktabs` package.²

A specific document format can be chosen by specifying the format’s name as an option to `esttab`. For example,

```
. esttab, tab
      (output omitted)
```

produces a tab-delimited table. As indicated above, the `smcl` mode is the default. However, if `using filename` is specified, the default format depends on `filename`’s suffix (e.g. `rtf` for “.rtf”, `html` for “.htm” or “.html”). Furthermore, if `filename` is specified without suffix, a default suffix is added depending on the specified document format (e.g. “.tex” for `tex` or `booktabs`).

2. See <http://www.ctan.org/tex-archive/macros/latex/contrib/booktabs/>.

Use with Excel

To produce a table for use with Excel, apply the `csv` format (or the `scsv` format depending on the language version of Excel). For example, type

```
. esttab using example.csv  
(output written to example.csv)
```

and then click on “example.csv” in Stata’s results window to launch Excel and display the file’s content.

Note that, depending on whether the `plain` option is specified or not, `esttab` uses two different variants of the CSV format. By default, that is, if `plain` is omitted, the contents of the table cells are enclosed in double quotes preceded by an equal sign (i.e. `=“...”`). This prevents Excel from trying to interpret the contents of the cells and, therefore, preserves formatting elements such as parentheses around t statistics. One drawback of this approach is, however, that the displayed numbers cannot directly be used for further calculations in Excel. Hence, if the purpose of exporting the estimates is to do additional computations in Excel, specify the `plain` option. In this case, the table cells are enclosed in double quotes without the equal sign, and Excel will interpret the contents as numbers.

Use with Word

To produce a table for use with Word, apply the `rtf` format. For example, type

```
. esttab using example.rtf  
(output written to example.rtf)
```

and then click on “example.rtf” in Stata’s results window to launch Word (or another RTF Reader, depending on your operating system settings) and display the table. Note that you may use `estout`’s `varwidth(#)` and `modelwidth(#)` options to change the column widths in the RTF table (`# = 12` corresponds to a column width of one inch, save cell padding). Furthermore, you may use the `append` option to include several tables in one RTF document.

Use with L^AT_EX

Using `esttab` together with L^AT_EX can be very effective. For example, Table 1 in this document has been produced by running the command

```
. esttab using example1.tex, label nostar title(Regression table\label{tab1})  
(output written to example1.tex)
```

and including

```
\input{example1.tex}
```

in the L^AT_EX code.

Table 1: Regression table

	(1)	(2)
	Price	Price
Weight (lbs.)	1.747 (2.72)	3.465 (5.49)
Mileage (mpg)	-49.51 (-0.57)	21.85 (0.29)
Car type		3673.1 (5.37)
Constant	1946.1 (0.54)	-5853.7 (-1.73)
Observations	74	74

t statistics in parentheses

Note that `esttab` automatically initializes the tabular environment and, if `title()` is specified, sets the table as a float object. Use the `fragment` option if you prefer to hard-code the table's environment and have `esttab` just produce the table rows.

Table 1 looks alright, but there is room for improvement. For example, the vertical spaces after the horizontal lines seem too small, the vertical gaps between the coefficients appear too large, and the use of double lines is debatable. One remedy for these problems is to load L^AT_EX's `booktabs` package in the document preamble and choose the `booktabs` format in `esttab`.

Other improvements to Table 1 would be to use a typographical minus sign and to align the numbers on the decimal point. These improvements can be implemented, for example, by loading L^AT_EX's `dcolumn` package³ and using `esttab`'s `alignment()` option to set a D column specifier. For instance, a good result might be attained by

```
. esttab using example2.tex, booktabs alignment(D{.}{.}{-1})
(output written to example2.tex)
```

Last but not least, it can be desirable to space out the table columns to a certain total table width. This is achieved using `esttab`'s `width()` option. For example, type

```
. esttab using example3.tex, width(0.6\hsize)
(output written to example3.tex)
```

to set the table width to 60% of the width of the text body and add white space between the columns.

3. See <http://www.ctan.org/tex-archive/macros/latex/required/tools/>.

Viewing the internal estout call

Sometimes, a desired table cannot be produced with standard `esttab` options right away. One approach in such cases is to use `esttab` to assemble a table that comes close, and then hand-edit and re-run the `estout` call that has been compiled by `esttab`. The `estout` call can be made visible by the `noisily` option and is also returned in `r(estout)`. Example:

```
. esttab, noisily
estout ,
  cells(b(fmt(a3) star) t(fmt(2) par("{ralign 12:{txt:}" "{txt:}}")))
  stats(N, fmt(%18.0g) labels("N"))
  starlevels(* 0.05 ** 0.01 *** 0.001)
  varwidth(12)
  modelwidth(12)
  abbrev
  delimiter(" ")
  smcltags
  prehead("{hline @width}")
  posthead("{hline @width}")
  prefoot("{hline @width}")
  postfoot("{hline @width}" "t statistics in parentheses" @starlegend)
  varlabels(, end(" ") nolast)
  mlabels(, depvar)
  numbers
  collabels(, none)
  eqlabels(, begin("{hline @width}" " ") nofirst)
  level(95)
(output omitted)
```

4 Appendix

4.1 Syntax of `eststo`

```
[_]eststo [ name ] [ , options ] [ : command ]
```

```
[_]eststo drop { # | name } [ { # | name } ... ]
```

```
[_]eststo clear
```

where *name* must not be `drop` or `clear` and *command* is any command returning its results in `e()` (see [U] **26 Overview of Stata estimation commands**). `by` is allowed, if `eststo` is used as a prefix command (see [D] `by`). A brief list of `eststo`'s options is provided below. See the online help for details.

<i>options</i>	description
<code>[no]esample</code>	do not/do store <code>e(sample)</code> (default is <code>esample</code> in <code>eststo</code> and <code>noesample</code> in <code>_eststo</code>)
<code>title(string)</code>	specify a title for the stored estimation set
<code>addscalars(name exp [...] [, replace])</code>	

<code>refresh[(#)]</code>	add scalars to the stored estimation set
<code>nocopy</code>	overwrite a previously stored estimation set
<code>missing</code>	clear <code>e()</code> after storing the estimation set
	use missing values in the by groups

4.2 Syntax of `esttab`

`esttab` [*namelist*] [`using filename`] [*, options*]

where *namelist* is either `_all` or *name* [*name ...*], and *name* is the name of a stored estimation set. The `*` and `?` wildcards may be used in *namelist* and the results estimated last may be indicated by a period (`.`) even if they have not yet been stored. A brief list of `esttab`'s options is provided below. See the online help for details.

<i>options</i>	description
Main	
<code>b(fmt)</code>	specify format for point estimates
<code>beta[(fmt)]</code>	display beta coefficients instead of point estimates
<code>main(name [fmt])</code>	display statistics contained in <code>e(name)</code> instead of point estimates
<code>t(fmt)</code>	specify format for <i>t</i> statistics
<code>abs</code>	use absolute value of <i>t</i> statistics
<code>not</code>	suppress <i>t</i> statistics
<code>se[(fmt)]</code>	display standard errors instead of <i>t</i> statistics
<code>p[(fmt)]</code>	display <i>p</i> -values instead of <i>t</i> statistics
<code>ci[(fmt)]</code>	display confidence intervals instead of <i>t</i> statistics
<code>aux(name [fmt])</code>	display statistics contained in <code>e(name)</code> instead of <i>t</i> statistics
<code>[no]constant</code>	do not/do report the intercept
Significance stars	
<code>[no]star[sym # [...]]</code>	do not/do report significance stars and, optionally, redefine significance symbols and thresholds
<code>staraux</code>	attach stars to <i>t</i> statistics instead of point estimates
Summary statistics	
<code>r2[.], ar2[.], pr2[(fmt)]</code>	display raw, adjusted, or pseudo <i>R</i> -squared
<code>aic[(fmt)], bic[(fmt)]</code>	display Akaike's or Schwarz's information criterion
<code>scalars(scalarlist)</code>	display any other scalars contained in <code>e()</code>
<code>sfmt(fmtlist)</code>	set format(s) for <i>scalarlist</i>
<code>noobs</code>	do not display the number of observations
<code>obslast</code>	place the number of observations last

Layout

<code>wide</code>	place point estimates and t statistics beside one another
<code>[no]parentheses</code>	do not/do print parentheses around t statistics
<code>brackets</code>	use brackets instead of parentheses
<code>[no]gaps</code>	suppress/add vertical spacing
<code>[no]lines</code>	suppress/add horizontal lines
<code>noeqlines</code>	suppress lines between equations
<code>compress</code>	reduce horizontal spacing
<code>plain</code>	produce a minimally formatted table

Labeling

<code>label</code>	make use of variable labels
<code>title(string)</code>	specify a title for the table
<code>mtitles(strlist)</code>	specify model titles to appear in table header
<code>nomtitles</code>	disable model titles
<code>[no]depvars</code>	do not/do print dependent variables in header
<code>[no]numbers</code>	do not/do print model numbers in table header
<code>coeflabels(strlist)</code>	specify labels for coefficients
<code>[no]notes</code>	suppress/add notes in the table footer
<code>addnotes(strlist)</code>	add lines at the end of the table

Document format

<code>smcl</code> <code>fixed</code> <code>tab</code> <code>csv</code> <code>scsv</code> <code>rtf</code> <code>html</code> <code>tex</code> <code>booktabs</code>	set the document format
<code>fragment</code>	suppress table opening and closing (L ^A T _E X, HTML)
<code>page[(packages)]</code>	add page opening and closing (L ^A T _E X, HTML)
<code>alignment(string)</code>	set alignment within columns (L ^A T _E X, HTML, RTF)
<code>width(string)</code>	set width of table (L ^A T _E X, HTML)

Output

<code>replace</code>	overwrite an existing file
<code>append</code>	append the output to an existing file
<code>type</code>	force printing the table in the results window
<code>noisily</code>	display the executed <code>estout</code> command

Advanced

<code>drop(droplist)</code>	drop individual coefficients or equations
<code>keep(keeplist)</code>	keep individual coefficients or equations
<code>order(orderlist)</code>	change order of coefficients and equations
<code>equations(matchlist)</code>	match the models' equations
<code>eform</code>	report exponentiated coefficients
<code>margin</code>	report marginal effects or elasticities
<code>unstack</code>	place multiple equations in separate columns
<code>other_estout_options</code>	any other <code>estout</code> options ⁴

4.3 Changes to `estout`

Numerous changes have been made to `estout` since its first publication in Jann (2005). Some of the more important changes are:

- `estout`'s `fmt()` suboption (within `cells()` and `stats()`) now provides two alternatives to official Stata's hard-to-type display formats. A fixed display format may now be specified as a single integer indicating the number of decimal places to be displayed. Furthermore, formats may now also be specified as `a#`, where `#` is in $\{1, 2, \dots, 9\}$. This causes `estout` to choose a reasonable format for each number depending on its scale. `#` sets the minimum number of "significant digits" to be displayed (see `help estout`, `marker(fmt)` for details).
- `style(smcl)` now produces SMCL formatted tables for display in Stata's results window.
- The `*` and `?` wildcards are now allowed in coefficient and equation specifications within options such as `drop()` and `keep()` and there is a new `order()` option to change the order of the coefficients in the table. In turn, `keep()` does not alter the order of the coefficients.
- The new `indicate()` option indicates for each model whether certain variables are present in the model. For example, if some of the models contain a set of year dummies, say `y1`, `y2`, and `y3`, you may specify

```
. estout ..., indicate(year effects = y1 y2 y3)
```

to drop the dummies from the table and add a row indicating for each model whether the year dummies are part of it or not.

- The new `refcat()` option inserts information on the reference category of a categorical variable in the model.
- The new `transform()` option applies transformations to coefficients, standard errors and confidence intervals. `transform()` is a generalization of the `eform` option and allows you, for example, to apply different kinds of transformations to different coefficients (say, apply exponentiation to the random effects part of a `xtmixed` model, but leave the rest unchanged).

4. All `estout` options are allowed in `esttab`. However, if specified, `estout` options take precedence over `esttab` options. For example, using `estout`'s `cells()` option will disable a whole series of `esttab` options (`b()`, `t()`, `abs`, `not`, `se()`, `p()`, `ci()`, `aux()`, `beta()`, `star`, `staraux`, `parentheses`, and `brackets`, to be precise). Furthermore, `estout`'s `stats()` option disables `r2()`, `ar2()`, `pr2()`, `aic()`, `bic()`, `scalars()`, `sfmt()`, `noobs`, and `obslast`. Other `estout` options that should be used with care are `begin()`, `delimiter()`, `end()`, `prehead()`, `posthead()`, `prefoot()`, `postfoot()`, `mlabels()`, and `varlabels()`.

- **estout** now takes action to clean up the table if equation names are different from model to model. In Stata 9, many commands return results using multiple equations, which often disarranges the table. **estout** now automatically matches first equations, if the equation names are different.
- Specifying **eqlabels(,none)** now causes **_cons** to be replaced by the equation name, if **_cons** is the only parameter in an equation. This is useful, for example, for tabulating **ologit** or **oprobit** results in Stata 9, which return the cut values as single equations containing just a constant.

4.4 Revision of **estadd**

estadd, also introduced by Jann (2005), has a new and simplified syntax and its functionality has been extended. The syntax, now similar to official Stata's **estat** command (available since Stata Version 9), is:

```
estadd subcommand [ , options ] [ : namelist ]
```

where *namelist* consists of names of stored estimation sets. If *namelist* is empty, **estadd** is applied to the last (i.e. currently active) estimates. Options are:

<i>options</i>	description
replace	permit estadd to overwrite existing e() results
prefix(string)	specify a common prefix for names of added results
subcmdopts	specific options associated with <i>subcommand</i> (see the online documentation)

estadd has three kinds of subcommands. First, there are elementary functions to add a simple macro, scalar, or matrix to the **e()**-returns. For example, use the **scalar** subcommand to add results from **test**:

```
. regress price weight mpg
(output omitted)
. test weight = mpg
(output omitted)
. estadd scalar p_diff = r(p)
```

The second type comprises subcommands that compute and add additional statistics for each coefficient in the model. For example, the **beta** subcommand adds a vector containing standardized beta coefficients, and the **mean** subcommand adds a vector containing the means of the regressors. Once added, these statistics can be tabulated in **estout** using the **cells()** option. Example:

```
. regress price weight mpg
(output omitted)
. estadd mean
```

```
. estout, cells("b mean") style(fixed)
```

	b	mean
weight	1.746559	3019.459
mpg	-49.51222	21.2973
_cons	1946.069	

The third type contains subcommands to compute and add scalar summary statistics that can then be tabulated in `estout` using the `stats()` option.

A brief list of the available subcommands is provided below. See the online help for details.

<i>subcommands</i>	<i>description</i>
Elementary	
<code>local name ...</code>	add a macro
<code>scalar name = exp</code>	add a scalar
<code>matrix name = mat [, copy]</code>	add a matrix
Statistics for each coefficient	
<code>beta</code>	standardized coefficients
<code>vif [, tolerance <u>sqr</u>vif]</code>	variance inflation factors (after <code>regress</code>)
<code>pcorr [, semi]</code>	partial (and semi-partial) correlations
<code>expb [, <u>no</u>constant]</code>	exponentiated coefficients
<code>ebsd</code>	standardized factor change coefficients
<code>mean</code>	means of regressors
<code>sd [, <u>no</u>binary]</code>	standard deviations of regressors
<code>summ [, stats]</code>	various descriptives of the regressors; the default <i>stats</i> are <u>mean</u> , <u>sd</u> , <u>min</u> , and <u>max</u> ; further <i>stats</i> are <u>sum</u> , <u>range</u> , <u>var</u> , <u>cv</u> , <u>semean</u> , <u>skewness</u> , <u>kurtosis</u> , <u>p1</u> , <u>p5</u> , <u>p10</u> , <u>p25</u> , <u>p50</u> , <u>p75</u> , <u>p90</u> , <u>p95</u> , <u>p99</u> , <u>iqr</u> , <u>all</u> , <u>median</u> , and <u>q</u>
Summary statistics	
<code>coxsnell</code>	Cox and Snell's pseudo <i>R</i> -squared
<code>nagelkerke</code>	Nagelkerke's pseudo <i>R</i> -squared
<code>lrtest model [, options]</code>	likelihood-ratio test; <i>options</i> are <u>name</u> (<i>string</i>) and <i>lrtest_options</i>
<code>ysumm [, stats]</code>	descriptives of the dependent variable; <i>stats</i> are as for the <code>summ</code> subcommand

5 Acknowledgments

`esttab` and `estout` owe much to John Luke Gallup's `outreg` (Gallup 1998) and official Stata's `estimates table` (see [R] `estimates`). Furthermore, the idea to provide the adaptive display format was motivated by Roy Wada's `outreg2`.

Numerous people commented on `estout` and reported bugs. Kit Baum, Debra Hevenstone, and J. Scott Long made comments on this article. I would like to thank them all.

6 References

- Burke, S. M. 2003. *RTF Pocket Guide*. O'Reilly Media.
- Gallup, J. L. 1998. sg97: Formatting regression output for published tables. *Stata Technical Bulletin* 46: 28–30.
- Jann, B. 2005. Making regression tables from stored estimates. *The Stata Journal* 5(3): 288–308.